



ODS GRAPHICS DESIGNER (Creating Templates for Batchable Graphs)

*Golden Horseshoe SAS User Group
October 14, 2011*

Barry Hong
BYHong@uss.com



History of SAS Graphics

- ***In the beginning there was PROC PLOT***
Crude raster graphics in the days of line printers
- ***Then there was SAS/GRAPH and it was better***
Vector graphics produced quality output
AXIS, FOOTNOTE, GOPTIONS, LEGEND, PATTERN, SYMBOL, TITLE
Lots of options but too many to learn effectively
Output stored in graphics catalogs
Not too friendly with Microsoft Office products
- ***SG Graphics***
Output as PNG file for sharing with Microsoft Office products.
Still code driven but using a new language employing styles
- ***Graphics Template Language***
Quality graphics fully compatible with Word and PowerPoint




- ***Eliminates the need to create template styles for graphics***
Who really mastered PROC TEMPLATE?
- ***Drag & Drop and Point & Click version of SG Graphics***
Let ODS Graphics Designer write the code for you
Customize the appearance to meet corporate standards
- ***Create Custom Designed Graphics***
Can layer charts or create panels in one file.
Make those Excel lovers jealous!
- ***Save the Template for Reuse or Sharing***
PROC SGRENDER processes data through the template
- ***Preproduction in SAS 9.2 Release 2 (TS2M0)***
- ***Part of Base SAS in 9.3 (No need for SAS/GRAPH)***



ODS Graphics Designer - Example Output


*Combination
bar chart by
month and
by week*





ODS Graphics Designer - Example Output

**Combination
scatter plot
and histogram
for the same
data.**



JOB ID

Footnote




ODS Graphics Designer - Example Output

**Combination
box plot by
month and 3
line charts
overlaid by
day for the
past 30 days.**

**Let's build the
template in
ODS
Graphics
Designer**

%sgdesign;





Complex templates may require a common data table

- **Chart_1_data (for box plot)**

<i>Month</i>	<i>Flow</i>
<i>JUN10</i>	<i>36.3</i>
<i>JUN10</i>	<i>32.4</i>
<i>...</i>	<i>...</i>
<i>MAY11</i>	<i>36.5</i>

```
DATA data.Chart_Data;  
set chart_1_data chart_2_data;  
RUN;
```

- **Chart_2_data (for line charts)**

<i>Date</i>	<i>Min</i>	<i>Mean</i>	<i>Max</i>
<i>11APR1123</i>	<i>36</i>	<i>40</i>	
<i>12APR11</i>	<i>30</i>	<i>30</i>	<i>30</i>
<i>...</i>	<i>...</i>	<i>...</i>	<i>...</i>
<i>10MAY11</i>	<i>12</i>	<i>38</i>	<i>41</i>



ODS Graphics Designer – Graph Gallery

Have It Your Way, Select a Graph from the Gallery

The screenshot displays the ODS Graphics Designer - Graph Gallery interface. The main window is titled "ODS Graphics Designer - Graph Gallery" and features a menu bar (File, Edit, View, Insert, Format, Tools, Help) and a toolbar. On the left, the "Elements" panel is open, showing a grid of "Plot Layers" and "Insets".

Plot Layers:

- Scatter, Series, Needle, Step
- Histogram, Histogram(H), Box, Box(H)
- Bar, Bar(H), BarError, BarError(H)
- Band, Vector, Contour, Fringe
- Normal, Normal(H), Kernel, Kernel(H)
- Loess, Regression, PBSpline, Ellipse
- Ref(H), Ref(V), DropLine, Line
- Block, StackBlock

Insets:

- Discrete Legend, Cell Header, Text Entry, Gradient Legend

The "Graph Gallery" window is open, showing a grid of graph types under the "Basic" tab. The gallery includes:

- ScatterPlot
- SeriesPlot
- StepPlot
- Histogram
- Vertical Box
- Horizontal Box
- Vertical Bar
- Horizontal Bar
- ContourPlot

At the bottom of the Graph Gallery window, there are buttons for "Organize", "Properties...", "OK", and "Close".




The Graph Gallery contains a tabbed set of commonly used graphs, organized as follows:

- **Basic** – Common graphs
- **Grouped** – Graphs showing grouped data
- **Analytical** – Graphs commonly used for analysis of data
- **Custom** – A set of graphs showing the possible ways to combine the plots
- **Matrix** – A set of Scatter Plot Matrix graphs
- **Panels** – A set of Classification Panel Graphs
- **MyGraphs** – A user defined group



ODS Graphics Designer – Assign Data


When you select a graph the Assign Data dialog box opens. Select the data table and the variables for the vertical box plot.





Default settings for the vertical box plot.

To create a second chart: Right mouse click on the chart and select Add Column.





ODS Graphics Designer – Plot Layers


Drag and drop a Series chart from the Plot Layers onto the new plot space.

The screenshot shows the ODS Graphics Designer interface. The title bar reads "ODS Graphics Designer - Graph - Graph Size = (680, 480)". The menu bar includes "File", "Edit", "View", "Insert", "Format", "Tools", and "Help". Below the menu bar is a toolbar with various icons. The "Elements" panel is open, showing a grid of "Plot Layers". The "Series" icon, which depicts a line graph, is circled in blue. A blue arrow points from this icon to a large dashed rectangular area in the "Graph" window. This area is labeled "Type in your title..." at the top and "Type in your footnote..." at the bottom. Inside this area, a box plot is displayed with the y-axis labeled "U/min" (ranging from 20 to 40) and the x-axis labeled "month" (ranging from Jun10 to May11). To the right of the dashed area, the text "(drop a plot here...)" is displayed.



ODS Graphics Designer – Plot Layers


The same data table for the box plot is used for the Series plot. Assign DATE and MEAN to the X & Y variables.





ODS Graphics Designer – Plot Layers


Repeat the Series Plot Layer drag and drop process on top of the Series Plot to add the MIN and MAX.





Now that we have created the basic design it is time to customize the appearance.


Select a chart component and right mouse click to change the properties.





ODS Graphics Designer – Customization

Select either chart, bring up the Graph Properties and select Common Row Axis for both charts to use the same Y axis scaling.







ODS Graphics Designer – Customize Plot Properties

**Unselect
Outline for
each of the
plots.**


**For each of
the series
select a
colour and
make the line
heavier
Thickness=2**






ODS Graphics Designer – Customize Plot Properties

Axes Tab permits colour, font and size changes to values and labels. Make changes to each axis separately.



Display Tab allows you to turn Label, Values, Grid and Tick Marks off and on.



**Albany AMT is the new Arial.
Font size unit is Pts**




ODS Graphics Designer – Customize

The box plot was widened by dragging the right border of the plot.

Title, ylabel & footnote made generic.


Next we want to drag and drop a legend onto the Series chart.






We will move the legend when we look at the code.

ODS Graphics Designer is built with only a subset of the Template code





Now to use the PROC TEMPLATE Code






ODS Graphics Designer – Understanding the Code

Select View > Code. In the view window copy all of the code and paste it into the SAS Editor.

We will modify the code to create a template for future use.

```
proc template;
  define statgraph sgdesign;
    dynamic _FLOW _MONTH _DATE _DATE2 _MIN _DATE3 _MAX _MEAN;
    begingraph / designheight=480 designwidth=720;
    entrytitle _id='title' halign=center '_TITLE' / textattrs=(size=14 family='Albany AMT');
    entryfootnote _id='footnote' halign=left '_FOOTNOTE' /;
    layout lattice _id='lattice' / columndatarange=data columngutter=10 columnweights=(0.5252365930599369
      0.47476340694006314 ) columns=2 rowdatarange=union rowgutter=10;
    layout overlay _id='overlay' / walldisplay=(FILL) xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL
      size=10 family='Albany AMT') display=(TICKS TICKVALUES LINE ));
    boxplot _id='box' x=_MONTH y=_FLOW / name='box';
    endlayout;
    layout overlay _id='overlay2' / walldisplay=(FILL) xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL
      size=10 family='Albany AMT') display=(TICKS TICKVALUES LINE ));
    seriesplot _id='series' x=_DATE y=_MEAN / connectorder=xaxis lineattrs=(thickness=2) name='series';
    seriesplot _id='series2' x=_DATE2 y=_MIN / connectorder=xaxis lineattrs=(color=CX9C3418
      thickness=2) name='series2';
    seriesplot _id='series3' x=_DATE3 y=_MAX / connectorder=xaxis lineattrs=(color=CX0000FF
      thickness=2) name='series3';
    discretelegend _id='legend' 'series' 'series2' 'series3' / border=true displayclipped=true down=1
      halign=center location=inside opaque=false order=columnmajor valign=bottom;
    endlayout;
  rowaxes;
    rowaxis _id='rowaxis' / label='_YLABEL' labelattrs=(weight=BOLD style=NORMAL family='Albany AMT')
      tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT');
  endrowaxes;
  endlayout;
endgraph;
end;
```

run;





Change name

Dynamic like macro

arguments: all start with underscore.

Consolidate and make generic.

Remove

quotes from

***TITLE** and*

***FOOTNOTE**
(now dynamic variables)*

```
proc template;  
  define statgraph sgdesign;  
  dynamic FLOW MONTH DATE DATE2 MIN DATE3 MAX MEAN;  
  begingraph / designheight=480 designwidth=720;  
    entrytitle _id='title' halign=center 'TITLE' / textattrs=(size=14 family='Albany AMT');  
    entryfootnote _id='footnote' halign=left 'FOOTNOTE' /;
```

Modified code

```
proc template;  
  define statgraph SG_GHSUG;  
  
  dynamic _YVARBOX MONTH DATE MIN MAX MEAN TITLE FOOTNOTE  
          _YLABEL;  
  
  begingraph / designheight=480 designwidth=720;  
    entrytitle _id='title' halign=center TITLE / textattrs=(size=14 family='Albany AMT');  
    entryfootnote _id='footnote' halign=left FOOTNOTE /;
```



Layout lattice defines layout of charts.

Simplify columnweight and reduce column gutter

Layout overlay for the boxplot.

Replace boxplot Y variable

_FLOW** with dynamic **_YVARBOX

```
layout lattice _id='lattice' / columndatarange=data columngutter=10  
columnweights=(0.5252365930599369 0.47476340694006314 ) columns=2  
rowdatarange=union rowgutter=10;
```

```
layout overlay _id='overlay' / walldisplay=(FILL)  
axisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10  
family='Albany AMT') display=(TICKS TICKVALUES LINE ));  
boxplot _id='box' x=_MONTH y=_FLOW / name='box';  
endlayout;
```

Modified code

```
layout lattice _id='lattice' / columndatarange=data columngutter=5  
columnweights=(0.55 0.45 ) columns=2  
rowdatarange=union rowgutter=10;
```

```
layout overlay _id='overlay' / walldisplay=(FILL)  
axisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10  
family='Albany AMT') display=(TICKS TICKVALUES LINE ));  
boxplot _id='box' x=_MONTH y=_YVARBOX / name='box';  
endlayout;
```



**Replace
_DATE2 and
_DATE3 with
common
dynamic
_DATE.**

**Rename
'series' as
'series1'**

**Move legend
location from
inside to
outside.**

```
layout overlay _id='overlay2' / walldisplay=(FILL)
  xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT')
    display=(TICKS TICKVALUES LINE ));
  seriesplot _id='series' x=_DATE y=_MEAN / connectorder=xaxis
    lineattrs=(thickness=2) name='series';
  seriesplot _id='series2' x=_DATE2 y=_MIN / connectorder=xaxis
    lineattrs=(color=CX9C3418 thickness=2) name='series2';
  seriesplot _id='series3' x=_DATE3 y=_MAX / connectorder=xaxis
    lineattrs=(color=CX0000FF thickness=2) name='series3';
  discretelegend _id='legend' 'series' 'series2' 'series3' / border=true displayclipped=true
    down=1 halign=center location=inside opaque=false order=columnmajor
    valign=bottom;
endlayout;
```

Modified code

```
layout overlay _id='overlay2' / walldisplay=(FILL)
  xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT')
    display=(TICKS TICKVALUES LINE ));
  seriesplot _id='series1' x=_DATE y=_MEAN / connectorder=xaxis
    lineattrs=(thickness=2) name='series1';
  seriesplot _id='series2' x=_DATE y=_MIN / connectorder=xaxis
    lineattrs=(color=CX9C3418 thickness=2) name='series2';
  seriesplot _id='series3' x=_DATE y=_MAX / connectorder=xaxis
    lineattrs=(color=CX0000FF thickness=2) name='series3';
  discretelegend _id='legend' 'series1' 'series2' 'series3' / border=true
    displayclipped=true down=1 halign=center location=outside opaque=false
    order=columnmajor valign=bottom;
endlayout;
```



**Remove
quotes from
_YLABEL
(now a
dynamic
variable)**

```
rowaxes;  
  rowaxis _id='rowaxis' / label='_YLABEL'  
    labelattrs=(weight=BOLD style=NORMAL family='Albany AMT')  
    tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT');  
endrowaxes;
```

Modified code

```
rowaxes;  
  rowaxis _id='rowaxis' / label=_YLABEL  
    labelattrs=(weight=BOLD style=NORMAL family='Albany AMT')  
    tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT');  
endrowaxes;
```



The PROC does not identify a data table!

It is a generic template.

Submit the code.

```
proc template;
  define statgraph SG_GHSUG;
  dynamic _YVARBOX _MONTH _DATE _MIN _MAX _MEAN _TITLE _FOOTNOTE _YLABEL;
  begingraph / designheight=480 designwidth=720;
  entrytitle _id='title' halign=center _TITLE / textattrs=(size=14 family='Albany AMT');
  entryfootnote _id='footnote' halign=left _FOOTNOTE /;
  layout lattice _id='lattice' / columndatarange=data columngutter=5 columnweights=(0.55 0.45 ) columns=2
    rowdatarange=union rowgutter=10;
  layout overlay _id='overlay' / walldisplay=(FILL) xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL
    size=10 family='Albany AMT') display=(TICKS TICKVALUES LINE ));
  boxplot _id='box' x=_MONTH y=_YVARBOX / name='box';
  endlayout;
  layout overlay _id='overlay2' / walldisplay=(FILL) xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL
    size=10 family='Albany AMT') display=(TICKS TICKVALUES LINE ));
  seriesplot _id='series1' x=_DATE y=_MEAN / connectorder=xaxis lineattrs=(thickness=2) name='series1';
  seriesplot _id='series2' x=_DATE y=_MIN / connectorder=xaxis lineattrs=(color=CX9C3418
    thickness=2) name='series2';
  seriesplot _id='series3' x=_DATE y=_MAX / connectorder=xaxis lineattrs=(color=CX0000FF
    thickness=2) name='series3';
  discretelegend _id='legend' 'series1' 'series2' 'series3' / border=true displayclipped=true down=1
    halign=center location=outside opaque=false order=columnmajor valign=bottom;
  endlayout;
  rowaxes;
  rowaxis _id='rowaxis' / label=_YLABEL labelattrs=(weight=BOLD style=NORMAL family='Albany AMT')
    tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT');
  endrowaxes;
  endlayout;
  endgraph;
end;
run;
```



***STATGRAPH
template is
saved locally
in a SAS
itemstore.***

***Want to save
templates to
an itemstore
to share with
others.***

***ODS path
statement will
search
SG_TEMPLATE
first.***

NOTE: STATGRAPH 'Sg_ghsug' has been saved to: SASUSER.TEMPLAT
24 run;

NOTE: PROCEDURE TEMPLATE used (Total process time):

real time	0.60 seconds
cpu time	0.12 seconds

```
proc template;  
  define statgraph sg_ghsug /store=libref.SG_TEMPLATE;
```

ODS path (prepend) libref.SG_TEMPLATE (read);

```
ods path show;  
Current ODS PATH list is:
```

***Add statement to
autoexec file.***

1. LIBREF.SG_TEMPLATE(READ)
2. SASUSER.TEMPLAT(UPDATE)
3. SASHELP.TMPLMST(READ)



ODS LISTING
*specifies the
output
location.*

**PROC
SGRENDER**
*acts like a
macro
invocation to
process the
data.*

*Output
created as a
PNG file.*

```
ods listing gpath="e:\temporary" image_dpi=100;
```

```
proc sgrender data=data.chart_data template=sg_ghsug  
  object=ghsug;  
  dynamic _DATE="date" _MONTH="month"  
    _MEAN="mean" _MAX="max" _MIN="min"  
    _YVARBOX="flow" _YLABEL="L/min"  
    _TITLE="Flow" _FOOTNOTE="My SG chart";  
run;
```

```
ods listing;
```

NOTE: Listing image output written to e:\temporary\ghsug1.png.

NOTE: There were 326 observations read from the data set
DATA.CHART_DATA.


NOTE: PROCEDURE SGRENDER used (Total process time):

real time	0.29 seconds
cpu time	0.09 seconds

Default path: d:\Program Files\SAS\SASFoundation\9.2\



Chart looks good except for the vertical axis on the Series chart.





**Create
yaxisopts for
the boxplot
and series
plot
overlays.**

```
layout overlay _id='overlay' / walldisplay=(FILL)
  xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10
    family='Albany AMT') display=(TICKS TICKVALUES LINE ));
  yaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10
    family='Albany AMT')
    display=(TICKS TICKVALUES LINE )
    labelattrs=(weight=BOLD style=NORMAL family='Albany AMT')
    label=_YLABEL);
  boxplot _id='box' x=_MONTH y=_YVARBOX / name='box';
endlayout;
```

```
layout overlay _id='overlay2' / walldisplay=(FILL)
  xaxisopts=(tickvalueattrs=(weight=BOLD style=NORMAL size=10
    family='Albany AMT') display=(TICKS TICKVALUES LINE ))
  yaxisopts=(display=none);
```


**Delete the
rowaxes
statements.**

```
rowaxes;
  rowaxis _id='rowaxis' / label=_YLABEL
  labelattrs=(weight=BOLD style=NORMAL family='Albany AMT')
  tickvalueattrs=(weight=BOLD style=NORMAL size=10 family='Albany AMT');
endrowaxes;
```



The finished chart as a PNG file.

How do I share the produced graph with others?





PDF Advantages

- *Multiple charts available in a single file*
- *More than one chart can be displayed on a page*
- *Charts can be click, copy and paste into WORD or PowerPoint !!!*
- *Charts can be part of a comprehensive report*



ODS graphics on and specify the height and width of the output.

No need to set options device=sasprt;

With ODS noresults there is no need for object= on the PROC SGRENDER line.

```
options orientation=landscape nonumber nodate;
ods escapechar='^';
ods listing image_dpi=100 close;
ods noresults;
ods pdf file="e:\temporary\file.pdf" notoc columns=2 startpage=no;

/* Output is 4 graphs to a page in a 2x2 layout with an ods startpage
after each column */
ods graphics on /height=3.3in width=4.9in;

proc sgrender data=data.chart_data template=sg_ghsug;
  dynamic _DATE="date" _MONTH="month"
    _MEAN="mean" _MAX="max" _MIN="min"
    _YVARBOX="flow" _YLABEL="L/min"
    _TITLE="Flow" _FOOTNOTE="My SG chart";
run;


... more charts

ods pdf close;
ods listing;
ods results;
```



Default graph size produced for 8.5 x 11 paper in portrait mode

Effect of Output Size on Font Appearance






Effect of Output Size on Font Appearance

Graph size
3.3in high
4.9in wide

Font height is absolute points. Font appears larger compared to the chart size.






Using the ODS Graphics Designer to Create Your Own Templates. Philip R Holland. Paper 034-2010, SAS Global Forum 2010

ODS Graphics Designer An Interactive Tool for Creating Batchable Graphs. Sanjay Matange. NESUG 2009 Poster


SAS/GRAPH® 9.2: ODS Graphics Designer Help.
What's New in SAS 9.3 ODS Graphics Designer.
Support.sas.com



QUESTIONS?



www.dilbert.com scottadams@aol.com



5-10-07 © 2007 Scott Adams, Inc./Dist. by UFS, Inc.

